Lecture 10

FIR filters 2

Introduction

In the previous lecture we developed a design procedure for FIR lowpass and highpass filters. In this lecture we consider bandpass and bandstop filters.

Bandpass filter

The desired frequency response of a bandpass filter (Fig. 1) is

$$H_{d}(\omega) = \begin{cases} 1 & \omega_{l} \le |\omega| \le \omega_{h} \\ 0 & \text{otherwise} \end{cases}$$
(1)

where ω_l, ω_h are the low- and high-frequency limits of the pass band. In the previous lecture we showed that

$$h_d(n) = \frac{1}{\pi} \int_0^{\pi} H_d(\omega) \cos(\omega n) d\omega$$
(2)

and

$$\frac{1}{\pi} \int_{\omega_1}^{\omega_2} \cos(\omega n) d\omega = \frac{\sin(\omega_2 n) - \sin(\omega_1 n)}{\pi n}$$
(3)

For the special case n=0 the value is simply $(\omega_2 - \omega_1)/\pi$. Therefore, the impulse response of an ideal bandpass filter is

$$h_d(n) = \frac{\sin(\omega_h n) - \sin(\omega_l n)}{\pi n} , \ h_d(0) = \frac{\omega_h - \omega_l}{\pi}$$
(4)

As we did with lowpass and highpass filters, we convert this to a causal FIR filter by setting



Fig. 1: Four ideal filter frequency responses.

Fig. 2: Bandpass filter with $f_1=0.1$, $f_h=0.2$ and M=60. (Left) impulse response; (right) frequency magnitude response.

$$h(n) = \begin{cases} h_d(n - M/2) & 0 \le n \le M \\ 0 & \text{otherwise} \end{cases}$$
(5)

Example impulse and frequency responses are shown in Fig. 2.

We've seen that a highpass filter can be considered as the difference of an allpass filter and a lowpass filter with the same cutoff frequency

$$h_d^{HP}(n) = \delta(n) - h_d^{LP}(n)$$
(6)

A bandpass filter can be thought of as the difference of a lowpass filter with cutoff ω_h and a lowpass filter with cutoff ω_l

$$h_{d}^{BP}(n) = h_{d}^{LP,\omega_{h}}(n) - h_{d}^{LP,\omega_{h}}(n)$$
(7)

Bandstop filter

The desired frequency response of a bandstop filter is

$$H_{d}(\omega) = \begin{cases} 0 & \omega_{l} \le |\omega| \le \omega_{h} \\ 1 & \text{otherwise} \end{cases}$$
(8)

We can also write this as

$$H_{d}(\omega) = \begin{cases} 1 & 0 \le |\omega| \le \omega_{l} \\ 1 & \omega_{h} \le |\omega| \le \pi \\ 0 & \text{otherwise} \end{cases}$$
(9)

This represents the filter as the sum of a lowpass filter with cutoff frequency ω_l

$$h_d(n) = \frac{\sin(\omega_l n)}{\pi n}$$
, $h_d(0) = \frac{\omega_l}{\pi}$

FIR filters 2

and a highpass filter with cutoff frequency ω_h

$$h_d(n) = \frac{-\sin(\omega_h n)}{\pi n}$$
, $h_d(0) = 1 - \frac{\omega_h}{\pi}$

Therefore

$$h_d(n) = \frac{\sin(\omega_l n) - \sin(\omega_h n)}{\pi n} , \ h_d(0) = 1 - \frac{\omega_h - \omega_l}{\pi}$$
(10)

Comparing this to (4) we see that

$$h_d^{BS}(n) = \delta(n) - h_d^{BP}(n)$$
(11)

A bandstop filter is equivalent to the difference of an allpass filter and a bandpass filter.

Transfer function zeros

The transfer function of an FIR filter is

$$H(z) = b_0 + b_1 z^{-1} + b_2 z^{-2} + \dots + b_M z^{-M}$$
(12)

We can write this as

$$H(z) = z^{-M} b_0 \left[z^M + \frac{b_1}{b_0} z^{M-1} + \frac{b_2}{b_0} z^{M-2} + \dots + \frac{b_M}{b_0} \right]$$
(13)

The M^{th} order polynomial in brackets can be written in factored form

$$z^{M} + \frac{b_{1}}{b_{0}} z^{M-1} + \frac{b_{2}}{b_{0}} z^{M-2} + \dots + \frac{b_{M}}{b_{0}} = (z - \zeta_{1})(z - \zeta_{2}) \cdots (z - \zeta_{M})$$
(14)

Therefore

$$H(z) = b_0 z^{-M} (z - \zeta_1) (z - \zeta_2) \cdots (z - \zeta_M)$$
(15)

where ζ_k is the k^{th} root of the polynomial – the k^{th} zero of H(z). To within a multiplicative constant $|b_0|$, the frequency response magnitude is simply the product of distances from a point $e^{j\omega}$ on the unit circle to each of the zeros of H(z)

$$\left|\frac{H(e^{j\omega})}{b_0}\right| = \left|e^{j\omega} - \zeta_1\right| \left|e^{j\omega} - \zeta_2\right| \left|e^{j\omega} - \zeta_3 z^{-1}\right| \cdots \left|e^{j\omega} - \zeta_M\right|$$
(16)

This gives us a graphical way to think about FIR filter design (Fig. 3, Fig. 4) in terms of positioning the zeros of H(z) in the z plane. Our task is simply to choose M points in the z plane corresponding to the zeros of H(z). If any of these points has a non-zero imaginary component, then the complex conjugate point must also be present. Other than that, we are free to choose the points in any manner we wish.

One approach to filter design is to choose some criterion for the "quality" of the filter response, and then find the zeros that maximize this, using either analytic or numerical methods. An example is the Parks–McClellan algorithm [1].

Fig. 3: Lowpass filter frequency response magnitude $|H(\omega)|$ is $|b_0|$ times the product of distances from zeros to $e^{j\omega}$.

Representing a filter as quadratic factors

Note that we can express (15) in the alternate form

$$H(z) = b_0 (1 - \zeta_1 z^{-1}) (1 - \zeta_2 z^{-1}) (1 - \zeta_3 z^{-1}) \cdots (1 - \zeta_M z^{-1})$$
(17)

The ζ_k values are either real or complex. If two of them, say ζ_1, ζ_2 , are real then

$$(1 - \zeta_1 z^{-1})(1 - \zeta_2 z^{-1}) = 1 - (\zeta_1 + \zeta_2) z^{-1} + \zeta_1 \zeta_2 z^{-2}$$
(18)

Fig. 4: Bandpass filter frequency response magnitude $|H(\omega)|$ is $|b_0|$ times the product of distances from zeros to $e^{j\omega}$.

Scott Hudson

FIR filters 2

is a quadratic with real coefficients. If $\,\zeta_1,\zeta_2\,$ are complex conjugates, $\,\zeta_2\!=\!\zeta_1^*$, then

$$(1-\zeta_{1}z^{-1})(1-\zeta_{2}z^{-1})=1-(\zeta_{1}+\zeta_{2})z^{-1}+\zeta_{1}\zeta_{2}z^{-2}$$

=1-2 Re[ζ_{1}] $z^{-1}+|\zeta_{1}|^{2}z^{-2}$ (19)

and again we have a quadratic with real coefficients. It follows that any FIR filter transfer function can be expressed as factors of quadratics with real coefficients

$$H(z) = b_0 (b_0^{(1)} + b_1^{(1)} z^{-1} + b_2^{(1)} z^{-2}) (b_0^{(2)} + b_1^{(2)} z^{-1} + b_2^{(2)} z^{-2}) \cdots (b_0^{(K)} + b_1^{(K)} z^{-1} + b_2^{(K)} z^{-2})$$

where M = 2K or M = 2K - 1. In the second case (*M* is odd) one of the quadratics will be a linear term, say, $b_0^{(K)} + b_1^{(K)} z^{-1}$.

References

1. https://en.wikipedia.org/wiki/Parks%E2%80%93McClellan_filter_design_algorithm

Scilab code

```
//FIR lowpass impulse response with 2*K+1 samples
//and cutoff frequency fc. (K=M/2)
function h = hLP(fc, K)
 wc = 2*%pi*fc;
 h = zeros(1, 2*K+1);
 h(K+1) = wc/%pi;
 for n=1:K
   h(K+1+n) = sin(wc*n)/(%pi*n);
   h(K+1-n) = h(K+1+n);
 end
endfunction
//FIR highpass impulse response with 2*K+1 samples
//and cutoff frequency fc. (K=M/2)
function h = hHP(fc, K)
 h = -hLP(fc, K);
 h(K+1) = h(K+1)+1;
endfunction
//FIR bandpass impulse response with 2*K+1 samples
//and passband frequencies fl, fh. (K=M/2)
function h = hBP(fl, fh, K)
 wl = 2*%pi*fl;
 wh = 2*%pi*fh;
 h = zeros(1, 2*K+1);
 h(K+1) = (wh-wl)/%pi;
 for n=1:K
   h(K+1+n) = (sin(wh*n)-sin(wl*n))/(%pi*n);
   h(K+1-n) = h(K+1+n);
  end
endfunction
//FIR bandstop impulse response with 2*K+1 samples
//and stopband frequencies fl, fh. (K=M/2)
function h = hBS(fl, fh, M)
 h = -hBP(fl, fh, M);
 h(M+1) = h(M+1)+1;
endfunction
```